
Turn the Tables on Boring Reports with SAS® 9.2 and RTF Tagset Options
Louise S. Hadden, Abt Associates Inc., Cambridge, MA

ABSTRACT

SAS® 9.2 offers new ways to create presentation quality RTF documents directly from the Output Delivery System
(ODS). This paper highlights a new technique to rotate column headers in tabular RTF output without post-
processing in Microsoft Word or Open Office. Prior to SAS 9.2, this could only be accomplished using a post-
processing macro within SAS. Now, rotating column headers can easily be accomplished simply by specifying RTF
tagset OPTIONS. User-written special style functions will also be explored. Examples producing tables in 9.1.3
(using the post-processing macro and PROC TABULATE) and in 9.2 (using the RTF tagset and PROC TABULATE,
and the RTF destination and PROC REPORT) will be demonstrated. Full code samples are available from the
author upon request.

THE RTF TAGSET

ODS destinations such as RTF, HTML, and PDF arose from the desire to be able to route procedural output to
something other than the SAS listing. Each SAS procedure has its own “groups” of “objects” that are output. A
“template” had to be set up for each SAS procedure’s ODS output. These templates had to accommodate ODS
output directed toward different destinations, which are roughly analogous to file types. RTF output was designed to
be included in Microsoft Word® or other packages which can interpret RTF control strings.

As the output delivery system grew in popularity,
users began to request more control over the
destinations in order to create output that met
various regulatory standards, among other things.
SAS generally uses a “tagset”, or group of events, to
produce output going to various destinations (such
as RTF, HTML or PDF.) Eric Gebhart, Master of
ODS Markup, says “. . . a tagset is a template that
defines a collection of events, that ultimately defines
a destination. More simply, a tagset defines, or is, a
destination.” Originally, both RTF and PDF
destinations were (and still are) part of the PRINTER
tagset. SAS knows to interpret the ODS RTF
statement as “use the RTF specific part of the
PRINTER tagset to generate RTF output.”

RTF, like HTML, is an incredibly complex markup
language which contains formatting commands. An
RTF file can be viewed in a text editor as a text file.
(See screen shot to the left for an example.)
Microsoft has its own specification containing RTF
commands, which is massive and overwhelming in
scope. As you might guess, this specification

changes for each version of Microsoft Word. For Word 2003, the specification is 1.8. For Word 2007, the
specification is 1.9.1. You can download these specifications from the Microsoft download page.

The PRINTER TAGSET / RTF destination combination was and is useful, but in order to take RTF output into a new
dimension (i.e. out of the horizontal plane), a new RTF “measured” TAGSET was developed. The new RTF tagset
makes a number of things possible that were impossible before, and makes other things (like rotation of column
headers) a lot easier. It is a hybrid of the ODS PRINTER model and ODS MARKUP. SAS users can use the new
RTF tagset to more finely control their RTF output. Some examples of what can be accomplished using the RTF
tagset are: paragraph spacing, column header rotation, paneling, and repeated headers.

One difference between the PRINTER TAGSET / RTF destination and the RTF tagset that is immediately obvious to
the user is that it is no longer necessary to specify BODYTITLE on the RTF call to get your titles and footnotes in the

“body” of the text. The RTF tagset places titles and footnotes as tables in the RTF document and not as RTF
instructions to be placed by Microsoft Word. Similarly, SAS and the RTF tagset now determine where the page
breaks are for long tables and not Word. This means that: larger tables can be rendered since only a page of data
and its structures have to be held in memory at any one time; procedures like PROC TABULATE can recover row
headers over implicit page breaks and inform that the table continues on the next page; and merging the RTF tagset
docs into a master document will not replace the titles and footers from the master. Essentially, SAS (and by
extension the SAS user) is now in control of RTF output instead of allowing Microsoft Word to make the decisions.

There is nothing wrong with your television set.
Do not attempt to adjust the picture. We are
controlling transmission. If we wish to make it
louder, we will bring up the volume. If we wish to
make it softer, we will tune it to a whisper. We
will control the horizontal. We will control the
vertical. We can roll the image, make it flutter.
We can change the focus to a soft blur or
sharpen it to crystal clarity. For the next hour, sit
quietly and we will control all that you see and
hear. We repeat, there is nothing wrong with
your television set. You are about to participate
in a great adventure. You are about to
experience the awe and mystery which reaches
from the inner mind to...
The Outer Limits. — Opening narration – The
Control Voice – 1960s

Awe and mystery pretty much describe tagsets! Tagsets allow a better exploitation of the specific output type, and a
more flexible package for use and support. Using the flexibility of the new RTF (and other) TAGSET(S) allows the
creation of custom tagsets that fit users’ needs, including the ability to write style functions. The RTF tagset is simply
a PROC TEMPLATE job that can be modified or inherited from.

This paper will demonstrate the use of column header rotation and briefly touch upon the creation and use of a RTF
style function, but readers are encouraged to experience the awe and mystery and try out the other exciting new
features of the RTF tagset. For an example of customizing the RTF tagset for your particular needs, see Wayne
Hester’s paper “Teaching Your RTF tagset to Do Clever Tricks.”

DETAILS, DETAILS

I hope I have sold you on the attractions of the new RTF tagset, even before I show you the ease of rotating column
headers. There are a number of different resources available for finding out more. A listing of papers and SAS
documentation on RTF and TAGSETS are provided at the end of the paper. One of the easiest sources of
information, though, is in your ODS TAGSETS.RTF call. The RTF tagset has an OPTIONS statement, which we will
use later on to rotate our column headers. To find out more about the options available, include doc=”help” in your
OPTIONS statement (enclosed by parentheses) as shown below. This syntax also works for the HTML TAGSET.
You must specify ODS TAGSETS.RTF (or HTML) or specify the TAGSET as opposed to the destination the “long”
way, as in ODS MARKUP TAGSET=TAGSETS.RTF. The text associated with the TAGSET help file will be printed in
your log.

91 ods tagsets.rtf file="final_weighted_&basename..rtf"
92 OPTIONS(TRHDR="\trrh1750"
93 TROWHDRCELL="\cltxbtlr" doc="help") style=styles.minimal;
NOTE: Writing TAGSETS.RTF Body file: final_weighted_q41_info.rtf
==============================================================================
The RTF tagset Help Text.

This Tagset/Destination helps with the creation of RTF files
for MS Word
==============================================================================

These are the options currently supported by this tagset.

Sample usage:

ODS TAGSETS.RTF OPTIONS(doc='Quick',
CONTENTS='yes');

ODS TAGSETS.RTF OPTIONS(SPECIFIC_OPTION='value');

Debug_Level: No default value.
Usage: OPTIONS(Debug_Level=1)
Description:
Determine what debugging information should be printed
to the log. The values expected are numeric and can be used to
take whatever action is needed. Used in tagsets being debugged,
but requires a local tagset to be modified.

Doc: No default value.
Help: Displays introductory text and options.
Quick: Displays available options.
Settings: Displays Current settings.

CONTENTS: No default value.
Usage: OPTIONS(CONTENTS='yes')
Description:
Adds a table of contents page at the top of the file,
followed by a page break. This must occur before any other output
in order to have an effect.
'yes' and 'on' have the same action.

SECT: No default value.
Usage: OPTIONS(SECT='string')
Description:
Inserts RTF controls onto the section data specifications.
ODS tagsets.rtf OPTIONS(sect='string')
The special string 'NONE' prevents ANY section data from
being added.
Assigning 'no' or 'off' resets the option to NULL.

TABLES_OFF: No default value.
Usage: OPTIONS(TABLES_OFF='string')
Description:
When the string names a STYLE type, that style type is
not inserted into RTF table specs. The text is inserted
directly into the document.
To get a list of the style elements written to the SAS log
use the special qualifier:
OPTIONS(TABLES_OFF='STYLE_ELEMENTS').
Assigning 'no' or 'off' resets the option to NULL.

Example:
OPTIONS(TABLES_OFF='USERTEXT') will act on the TEXT='string'
since it uses the USERTEXT style. The text string will
not be placed in a table. The style name is case sensitive.

TOC_DATA: No default value.
Usage: OPTIONS(TOC_DATA='on')
Description:
Adds table of contents information to the file
A value of 'off' or 'no' stops the information from being added.

TROWD: No default value.
Usage: OPTIONS(TROWD='string')
Description:

Inserts raw RTF specifications directly onto the
table row header descriptions.
Assigning 'no' or 'off' resets the option to NULL.

TRHDR: No default value.
Current Value: \trrh1750
Usage: OPTIONS(TRHDR='string')
Description:
Inserts raw tablerow RTF specifications directly onto the
table row header descriptions.
Assigning 'no' or 'off' resets the option to NULL.

TROWHDRCELL: No default value.
Usage: OPTIONS(TROWHDRCELL='string')
Description:
Inserts raw text on the trow cells.
Assigning 'no' or 'off' resets the option to NULL.

OPTIONS TO PRODUCE ROTATED COLUMN HEADERS

THE OLD WAY: POST PROCESSING MACRO WRITTEN BY WAYNE HESTER OF SAS®

After attempting to accomplish my goal of rotating column headers unsuccessfully, I
enlisted the help of SAS Technical Support. I was referred to Wayne Hester of SAS®
and he had the answer with a neat little macro. It did involve post-processing, but, it was
post-processing in SAS®, within my reporting program. Instead of outputting directly to
the RTF destination, the SAS output is directed to an intermediate RTF file which is then
post-processed in a macro to change the RTF command for column header orientation
before being output into the final RTF file. The macro, while used to change the column
orientation in my case, could also be used to change other specifications.

RTF CONVERSION MACRO:

%macro do_one(source=xxx);
data temp ;
length line $400;
infile &source length=lg lrecl=1000 end=eof;

input @1 line $varying400. lg;
data _null_;

set temp ; retain mod 0 ;
file &source;

/* mods are done for the trow header data only */
if (index(line,"\trowd\trkeep") > 0) then do ;
if (index(line,"\trowd\trkeep\trhdr") > 0) then mod = 1;
else mod = 0;
end;
if (mod) then do;
/* this works because the commands are the same width */
line=tranwrd(line,"\cltxlrtb","\cltxbtlr"); line=tranwrd(line,"\clvertalb","\clvertalt");

line=tranwrd(line,"\qc","\ql") ;
put line;

/* The header cell needs to be tall enough to accept the change. */
/* You may need to adjust this depending on what your column headers are. */
/* Measurement is in twips. 20 twips to the point, 72 points to the inch.*/
if ( index(line,"\trowd\trkeep\trhdr") > 0) then put "\trrh-1625" ;

end;
else put line ;

run;
%mend;

%do_one(source=rtf) ;

The result was a perfectly sized landscape report with rotated column headers (clipped screen shot seen
below.)

THE NEW WAY: ODS TAGSETS.RTF OPTIONS

The two specific RTF tagset OPTIONS used to rotate column headers are TRHDR and TROWHDRCELL. TRHDR
allows you to feed in RTF commands relating to the table row header. In this case, we are giving the RTF tagset
information about the vertical height of the table row header, \trrh9999. The height is measured in twips; 20 twips to
the point, 72 points to the inch. You will want to play around with the number to get the height you need to
accommodate your column headers. For the example shown, I used 2750 twips. The TROWHDRCELL is where the
action occurs. We feed in the RTF command to produce a vertical label (\cltxbtlr) in the table row header cells. This
RTF command is the same one used in the RTF conversion macro. It’s just a lot easier to specify!

PROC TABULATE AND TAGSETS.RTF

/* note the trrh9999 may need to be adjusted based on length of labels for var
variables */

ods tagsets.rtf file="final_weighted_&basename..rtf"
OPTIONS(TRHDR="\trrh2750"

TROWHDRCELL="\cltxbtlr" ) style=styles.minimal;
ods escapechar='^';
proc tabulate data=anal. . .

var counter &colvar1 /
s={bordercolor=black just=l borderwidth=2 cellwidth=60
font=("Arial",8pt,Normal)};

PROC REPORT AND TAGSETS.RTF

ods tagsets.rtf file='dummy2.rtf' OPTIONS(TRHDR="\trrh1750"
TROWHDRCELL="\cltxbtlr" doc="help")

path=odsout style=styles.nmep;

ods escapechar='^';

proc report nowd data=page1 spacing=8
style(report)=[cellpadding=3pt vjust=b]
style(header)=[just=center font_face=Arial font_weight=bold font_size=8pt]
style(lines)=[just=left font_face=Arial] ls=132;

columns rowtitle wgtsum0 printvar1 printvar2 printvar3 printvar4 printvar5
printvar6 boldit;

define boldit / display ' ' noprint;
define rowtitle / style(COLUMN)={just=l font_face=Arial

font_size=8pt cellwidth=200 asis=on}
style(HEADER)={just=l font_face=Arial font_weight=bold

font_size=8pt cellwidth=200 asis=on};

define wgtsum0 / style(COLUMN)={just=c font_face=Arial background=lightgray
font_size=8pt celllwidth=80 asis=on}
style(HEADER)={just=l font_face=Arial font_weight=bold
background=lightgray font_size=8pt celllwidth=80 asis=on}; . . .

In order to left justify the headers (or bottom justify in this case) it is necessary to specify left justification. For the
PROC TABULATE example this is done with the VAR statement (highlighted in red in the code) and for the PROC
REPORT example it is done with the define style header statement.

It’s important to note that when resetting RTF tagset options (for example STARTPAGE=YES) that you use ODS
TAGSETS.RTF instead of ODS RTF. I learned this the hard way – using ODS RTF after ODS TAGSETS.RTF just
opens a new document using the ODS PRINTER / RTF destination, which I found very confusing. In addition, it is a
good practice to use the RTF extension for your output file instead of the shortcut of using the DOC extension, as
some RTF-specific functionality will be lost using the DOC extension.

ALTERNATIVES FOR SAVING SPACE

You might not want to rotate column headers if, for example, you were using spanning headers in PROC REPORT
or had a complicated PROC TABULATE which had the effect of spanning headers, because the text rotation occurs
for ALL header cells. There are alternatives.

PAPERSIZE=LEGAL / INLINE STYLES to increase space available and smaller font size (6pt).
ods listing close;
options orientation=landscape nodate nonumber nocenter papersize=legal;
ods rtf file='table1b_report.doc' path=odsout style=styles.journal bodytitle;
ods escapechar='^';

proc report nowd data=dd.merge_table1 spacing=8
style(report)=[cellpadding=3pt vjust=b]
style(header)=[just=center font_face=Arial font_weight=bold font_size=6pt]
style(lines)=[just=left font_face=Arial] ls=250 ;

USE ODS ESCAPECHAR n (linebreak) to “double up” columns.

data out&tabnum._&colnum (keep=country_order country_year_uc
printvar&tabnum._&colnum mean&tabnum._&colnum
semean&tabnum._&colnum);
length wgtsum meanse printvar&tabnum._&colnum $ 40;
set temp&tabnum&colnum tempa&tabnum&colnum (in=a);
colnum=&colnum;
tabnum=&tabnum;
semean=semean*100;
meanse=cats(put(mean,percent7.1),'^n#(',put(semean,7.2),')');
meanse=compress(meanse,'#');
wgtsum=put(count1,comma7.0);
printvar&tabnum._&colnum=cats(wgtsum,'^n',meanse);
mean&tabnum._&colnum=put(mean,percent7.1);
semean&tabnum._&colnum=meanse;

run;

ANOTHER REASON TO LOVE TAGSETS IF THE AWE AND MYSTERY AREN’T ENOUGH

As noted above, the RTF tagset offers greater flexibility than the ODS PRINTER / RTF destination in support and
features. One example of this is the ability to “inherit” from the RTF tagset using PROC TEMPLATE and write your
own style functions. Eric Gebhart of SAS has some examples of writing style functions for the HTML tagset, and an
example written by Wayne Hester of SAS for the RTF tagset to change font color based on word count follows. This
functionality allows the SAS user to create a customized tagset for specialized purposes.

proc template;
define tagset tagsets.bcrtf;
parent=tagsets.rtf;

define event bcol;

put nl;

put '{' ;
eval $count 1;
do /if index(value, ' ');

set $word scan(value, 1, ' ');
do /while !cmp($word, ' ');

do /if mod($count, 2);
put '\cf2 ' $word ' ';

else;
put '\cf4 ' $word ' ';

done;

eval $count $count + 1;
set $word scan(value, $count, ' ');

done;

else /if $event;
put '\cf1 ' value ' ';

done;

put '}' nl ;

end;

end;

run;

ods tagsets.bcrtf file="ex.rtf" ;

ods escapechar="^";

proc print data=sashelp.class noobs;
title1 ' Switch Colors i.e. ^{bcol switch colors } depending on word count';
title2 '^R/RTF"\cf2" DOTS DOTS DOTS';
title3 '^R/RTF"\cf4" THICK THICK THICK';
run;

ods _all_ close;

CONCLUSION

SAS and ODS RTF provide SAS programmers with tools to create attractive and informative reports. A new addition
to the ODS RTF family in version 9.2 is the RTF tagset. The RTF tagset contains many useful options, among them
the ability to rotate column headers without post-processing and the ability to write your own style functions.

The use of the ODS RTF output destination and the ODS RTF tagset allow the SAS programmer to produce complex
tables with highly controlled formatting that can be read directly into Microsoft Word. (While reading into OpenOffice
has also been tested, neither text rotation technique works with the current version of OpenOffice at the time of
writing of this paper.) The ease of table production using these methods greatly adds to the SAS programmer’s
toolbox.

It also cannot be emphasized enough that being able to reproduce professional, accurate output without error-prone
interventions such as copying and pasting is vital to the SAS programmer’s repertoire.

